

WA3 - CS 449 Summer 2019 – GDB and x86-64 Assembly

x86-64 Assembly Language

Assembly language is a human-readable representation of machine code instructions (generally a one-to-one correspondence). Assembly is machine-specific because the computer architecture and hardware are designed to execute a particular machine code instruction set.

x86-64 is the primary 64-bit instruction set architecture (ISA) used by modern personal computers. It was developed by Intel and AMD and its 32-bit predecessor is called IA32. x86-64 is designed for complex instruction set computing (CISC), generally meaning it contains a larger set of more versatile and more complex instructions.

For this course, we will utilize only a small subset of x86-64's instruction set and omit floating point instructions.

x86-64 Instructions

The subset of x86-64 instructions that we will use in this course take either one or two operands, usually in the form: `instruction operand1, operand2`. There are three options for operands:

- **Immediate:** constant integer data (*e.g.* `$0x400`, `$-533`) or an address/label (*e.g.* `Loop`, `main`)
- **Register:** use the data stored in one of the 16 general purpose registers or subsets (*e.g.* `%rax`, `%edi`)
- **Memory:** use the data at the memory address specified by the addressing mode `D(Rb, Ri, S)`

The operation determines the effect of the operands on the processor state and has a suffix (“b” for byte, “w” for word, “l” for long, “q” for quad word) that determines the bit width of the operation. Sometimes the operation size can be inferred from the operands, so the suffix is omitted for brevity.

Control Flow and Condition Codes

Internally, condition codes (Carry, Zero, Sign, Overflow) are set based on the result of the previous operation. The `je` and `set` families of instructions use the values of these “flags” to determine their effects. See the table provided on your reference sheet for equivalent conditionals.

An *indirect jump* is specified by adding an asterisk (*) in front of a memory operand and causes your program counter to load the address stored at the computed address. (e.g. `jmp *%rax`) This is useful for switch case statements

Procedure Basics

The instructions `push`, `pop`, `call`, and `ret` move the stack pointer (`%rsp`) automatically.

`%rax` is used for the return value and the first six arguments go in `%rdi`, `%rsi`, `%rdx`, `%rcx`, `%r8`, `%r9`
(“Diane’s Silk Dress Cost \$89”).

x86 instructions	English equivalent
<code>movq \$351, %rax</code>	Move the number 351 into 8-byte (quad) register “rax”
<code>addq %rdi, %rsi</code>	Add the 64-bit value of <code>%rdi</code> to <code>%rsi</code>
<code>movq (%rdi), %r8</code>	Move the 64-bit data at the address stored in <code>%rdi</code> to <code>%r8</code>
<code>leaq (%rax,%rax,8), %rax</code>	Compute <code>9 * %rax</code> , and store the 64-bit result in <code>%rax</code>

Exercises:

1. [CSE351 Au14 Midterm] Symbolically, what does the following code return?

```
movl  (%rdi), %eax      # %rdi -> x;  r = *x
leal  (%eax,%eax,2), %eax # %rax -> r;  r = (*x) * 3
addl  %eax, %eax        #              r = (*x)*3 + (*x)*3
andl  %esi, %eax        # %rsi -> y;  r = ((*x)*6) & y
subl  %esi, %eax        #              r = (((*x)*6) & y) - y
ret

(((*x) * 6) & y) - y
```

2. [CSE351 Au15 Midterm] Convert the following C function into x86-64 assembly code. You are not being judged on the efficiency of your code – just the correctness.

```
long happy(long *x, long y, long z) {
    if (y > z)
        return z + y;
    else
        return *x;
}
```

```
happy:
    cmpq  %rdx, %rsi
    jle   .else
    leaq  (%rdx, %rsi), %rax
    ret
.else:
    movq  (%rdi), %rax
    ret
```

Multiple other possibilities (*e.g.* switch ordering of if/else clauses, replace `lea` with `mov`/`add` instruction pair).

3. Write an equivalent C function for the following x86-64 code:

```
mystery:
1  testl    %edx, %edx          # %edx is 3rd argument (z)
2  js       .L3                # jump to .L3 if z<0
3  cmpl     %esi, %edx          # %esi is 2nd argument (y)
4  jge      .L3                # jump to .L3 if y>=z
5  movslq   %edx, %rdx          # sign-extend 3rd argument (z)
6  movl     (%rdi,%rdx,4), %eax  # %rdi is 1st argument (x), calc *(x + z*4)
7  ret
.L3:
8  movl     $0, %eax           # return 0
9  ret
```

```
int mystery(int *x, int y, int z) {
    if (z >= 0 && z < y)
        return x[z];
    else
        return 0;
}
```

Notes:

- If either conditional is True, then we jump to the “else” clause, so in C we execute the “if” clause only when the complement of both of them are True.
- Line 6 indicates that the return type is 4 bytes (int). Line 8 is ambiguous since it zeros out the entire 8 bytes of %rax.
- Argument variable names are arbitrary. Based on usage, could perhaps have used $x \rightarrow ar$, $y \rightarrow n$, $z \rightarrow k$.
- First argument had to point to int based on scale factor in Line 6. Both `int *x` and `int x[]` work.

4. [CSE351 Wi17 Midterm] Consider the following x86-64, (partially blank) C code, and memory diagram. Addresses and values are 64-bit. Fill in the C code based on the given assembly.

```
foo:
    movl    $0,    %eax

L1:
    testq   %rdi,   %rdi
    je      L2
    movq    (%rdi), %rdi
    addl    $1,     %eax
    jmp     L1

L2:
    ret
```

```
int foo(long* p) {
    int result = 0;
    while (p != NULL) {
        p = *(long**)p;
        result = result + 1;
    }
    return result;
}
```

Part 2: Follow the execution of foo in assembly, where 0x1000 is passed in to %rdi

Write the values of %rdi and %eax in the columns. If the value doesn't change, you can leave it blank

Instruction	%rdi (hex)	%eax (decimal)
movl	0x1000	0
testq		
je		
movq	0x1030	
addl		1
jmp		
testq		
je		
movq	0x0	
addl		2
jmp		
testq		
je		
ret		

Address	Value
0x1000	0x1030
0x1008	0x1020
0x1010	0x1000
0x1018	0x0000
0x1020	0x1030
0x1028	0x1008
0x1030	0x0000
0x1038	0x1038
0x1040	0x1048
0x1048	0x1040

Assembly Instructions

mov a, b	Copy from a to b.
movs a, b	Copy from a to b with sign extension. Needs two width specifiers.
movz a, b	Copy from a to b with zero extension. Needs two width specifiers.
lea a, b	Compute address and store in b. <i>Note: the scaling parameter of memory operands can only be 1, 2, 4, or 8.</i>
push src	Push src onto the stack and decrement stack pointer.
pop dst	Pop from the stack into dst and increment stack pointer.
call <func>	Push return address onto stack and jump to a procedure.
ret	Pop return address and jump there.
add a, b	Add from a to b and store in b (and sets flags).
sub a, b	Subtract a from b (compute b-a) and store in b (and sets flags).
imul a, b	Multiply a and b and store in b (and sets flags).
and a, b	Bitwise AND of a and b, store in b (and sets flags).
sar a, b	Shift value of b <i>right (arithmetic)</i> by a bits, store in b (and sets flags).
shr a, b	Shift value of b <i>right (logical)</i> by a bits, store in b (and sets flags).
shl a, b	Shift value of b <i>left</i> by a bits, store in b (and sets flags).
cmp a, b	Compare b with a (compute b-a and set condition codes based on result).
test a, b	Bitwise AND of a and b and set condition codes based on result.
jmp <label>	Unconditional jump to address.
j* <label>	Conditional jump based on condition codes (<i>more on next page</i>).
set* a	Set byte based on condition codes.